Abstract
We introduce the main themes, objectives and concepts of this class on Functional Programming. Without going into details, we mention the key ideas. We will return to these in more detail later. For now, our aim is to get you motivated and interested.

1 Themes
- functional programming
- correctness, termination, and performance
- types, specifications, and proofs
- evaluation and equivalence
- referential transparency and compositional reasoning
- exploiting parallelism

2 Objectives
- Learn to write functional programs
- Learn to write specifications
- Use rigorous techniques to prove program correctness
- Learn how to analyze the sequential and parallel running time of functional programs
- Choose data structures, design functions, and exploit parallelism to improve efficiency
- Structure your code using abstract types and modules, with clear and well designed interfaces
3 Functional Programming and ML

We use the programming language Standard ML, or just ML for short.

- ML is a functional programming language
  - computation = expression evaluation, to produce a value
- ML is a statically typed language
  - types are determined by a syntax-directed algorithm
  - there are simple and intuitive syntactic rules for designing well-typed expressions
- Evaluation is type-safe
  - well-typed expressions can be evaluated, and if evaluation terminates you get a value of the expected type
- ML is polymorphically typed
  - every well-typed expression has a most general type, and can be used safely at any instance of that type
- ML is a call-by-value language
  - function calls evaluate their arguments
- Functional programs are referentially transparent
  - expressions of the same type are “equivalent” when their evaluation produces equal results
  - compositional reasoning: substitution of “equals for equals”
- Functional programs are mathematical objects
  - can exploit mathematical techniques to prove correctness
- ML allows recursive definition of functions and datatypes
  - can use mathematical induction to analyze recursive code
- Evaluation order is largely irrelevant
  - safe to use parallel evaluation for independent code
  - strive to avoid unnecessary (re)-evaluation
4 Principles

- Expressions must be well-typed.
  Well-typed expressions don’t go wrong.

- Every function needs a specification.
  Well-specified programs are easier to understand.

- Every specification needs a proof.
  Well-proven programs do the right thing.

- Large programs should be designed as modules.
  Well-interfaced programs are easier to maintain and develop.

- Data structures algorithms.
  Sensible choice of data structure leads to better code.

- Think parallel, when feasible.
  Parallel programs may go faster.
  Use trees instead of lists if you can.

- Strive for simplicity.
  Programs should be as simple as possible, but no simpler.
  Simple code is usually easier to debug and easier to prove correct!

Some of these (all of them?) may sound trite or obvious to you, but (we intend to demonstrate and convince you that) these are fundamental and important principles that can improve your programming skills.
5 Today’s lecture

A brief intro to functional programming, with forward references to concepts, techniques and themes that will be developed later in the semester. Don’t worry about the details (of syntax, or of semantics!). For now, just try to appreciate the elegance and simplicity!

This write-up may not cover exactly what was shown in class. It serves to supplement the classroom slides and the material introduced during the lecture.

Types

Expressions in ML are typed; only well-typed expressions can be evaluated. This prevents many common programming errors!

For today, we will refer to the types int (integers), int list (lists of integers), int list list (lists of lists of integers) and function types like int list -> int (functions from integer lists to integers).

Functions

First, a (recursive) function for adding the integers in a list. Defined using cases: empty list, and non-empty list. The lines of form (* ... *) are comments and serve solely to document the intended type and specification of the function.

(* sum : int list -> int *)
fun sum [ ] = 0
  | sum (x::L) = x + (sum L)

(* Specification: *)
(* For all integer lists L, *)
(* sum(L) = the sum of the integers in L. *)

Note how similar this function definition is to a mathematical definition of a function. List notation: [ ] is the empty list, and [2,3] is the integer list with 2 as head and [3] as tail. Note that 1::[2,3]=[1,2,3]. We can use mathematical notation and math-style equational reasoning to explain how this function works:

\[
\text{sum } [1,2,3] \\
\text{ = 1 + sum } [2,3] \\
\text{ = 1 + (2 + sum } [3])
\]
Next, a function for adding the integers in a list of integer lists. Again recursive, again by cases. Uses \texttt{sum} from above.

\begin{verbatim}
(* count : (int list) list -> int *)
fun count [ ] = 0
  | count (r::R) = (sum r) + (count R)

(* Specification: *)
(* For all lists of integer lists R, *)
(* count R = the sum of the integers in the lists of R. *)
\end{verbatim}

Use equational reasoning to show that \texttt{count [[1], [2,3,4], [5]]} = 15.

**Evaluation**

Computation is evaluation. Expression evaluation stops when we reach a “value”, such as an integer numeral or a list of integer numerals. Addition expressions evaluate from left to right.

We write \( \Rightarrow^* \) for “evaluates in a finite number of steps to”.

\begin{verbatim}
(* sum [1,2,3] \Rightarrow^* 1 + sum [2,3] *)
(* \Rightarrow^* 1 + (2 + sum [3]) *)
(* \Rightarrow^* 1 + (2 + (3 + sum [ ])) *)
(* \Rightarrow^* 1 + (2 + (3 +0)) *)
(* \Rightarrow^* 1 + (2 + 3) *)
(* \Rightarrow^* 1 + 5 *)
(* \Rightarrow^* 6 *)
(* Numeral additions get done after recursive call returns. *)
\end{verbatim}

Similarly,

\begin{verbatim}
(* count [[1,2,3], [1,2,3]] \Rightarrow^* sum [1,2,3] + count [[1,2,3]] *)
(* \Rightarrow^* 6 + count [[1,2,3]] *)
(* \Rightarrow^* 6 + (sum [1,2,3] + count [ ]) *)
(* \Rightarrow^* 6 + (6 + count [ ]) *)
(* \Rightarrow^* 6 + (6 + 0) *)
(* \Rightarrow^* 6 + 6 \Rightarrow^* 12 *)
\end{verbatim}
Tail recursion

A recursive function definition is called *tail recursive* if in each clause of the function’s definition any recursive call is the last thing that gets done. The definitions of *sum* and *count* are not tail recursive, because for *sum* or *count* on a non-empty list there is an addition operation to do after the recursive call. You can see this in the layout of our evaluation display above.

Here is an addition function for integer lists that uses an extra argument as an *accumulator* to hold an integer representing the result of the additions so far, and does an addition onto the accumulator value before making the recursive call. The definition of *sum’* is tail recursive.

(* sum’ : int list * int -> int *)
fun sum’ ([ ], a) = a
| sum’ (x::L, a) = sum’ (L, x+a);

(* Specification: for all L:int list, a:int, sum'(L,a) = sum(L)+a. *)

Use equational reasoning to show that *sum’([1,2,3], 4) = 10.*

Later we will discuss tail recursion in more detail, in connection with efficient implementation; typically tail recursive functions need less stack space. Compare the shape of the evaluation diagram for *sum’* with that of *sum*, when applied to [1,2,3]. For the tail recursive version the shape doesn’t grow wide then shrink back.

(* sum’ ([1,2,3], 0) =>* sum’ ([2,3], 1) *)
(* =>* sum’ ([3], 3) *)
(* =>* sum’ ([ ], 6) *)
(* =>* 6 *)

Using *sum’* we can define a function *Sum* that is “equivalent” to *sum*:

(* Sum : int list -> int *)
fun Sum L = sum’ (L, 0);
Equivalence and Referential Transparency

It is easy to show by induction on the length of \( L \) that for all integer lists \( L \), \( \text{Sum} \ L = \text{sum} \ L \). Hence we say the functions \( \text{Sum} \) and \( \text{sum} \) are “extensionally equivalent”, or just “equivalent”, and we write \( \text{Sum} = \text{sum} \).

Induction is a key technique for proving correctness of recursive functions, and for proving termination. We will make extensive use of induction, in various and general forms, throughout the course.

A key property of functional programming is referential transparency, sometimes paraphrased as the assertion that in a functional programming language it is safe to replace “equals by equals”, or that “the value of an expression depends only on the values of its subexpressions”. Here we state a (slightly simplified, but more precise) version of this property.

First, for each type we define a notion of “equivalence”:

(* Expressions of type int are "equivalent" *)
(* if they evaluate to the same integer. *)

(* Expressions of type int list are "equivalent" *)
(* if they evaluate to the same integer list. *)

(* Function expressions are "equivalent" *)
(* if, when applied to "equivalent" arguments *)
(* they produce "equivalent" results. *)

Referential transparency is the property that: replacing a sub-expression by an “equivalent” sub-expression produces an “equivalent” expression.

WARNING: we limit attention here to “pure” functional programs. For the full ML language we need a more sophisticated version of referential transparency in which we take account of “effects” such as runtime errors and side-effects.

Examples

(* 21+21 and 42 are equivalent expressions of type int. *)
(* The expressions (21+21)*3 and 42*3 have the same value. *)

(* The functions Sum and sum, of type int list -> int, are equivalent. *)
Now consider the following function definition:

(* Count : int list list -> int *)
fun Count [ ] = 0
   | Count (r::R) = Sum r + Count R;

By referential transparency, it follows that Count is equivalent to count.

The above discussion is an example of compositional reasoning based on referential transparency. For the “pure” functional subset of ML, it is safe to replace an expression by an equivalent expression (of the same type). The result will be equivalent to the original program. We do this kind of compositional or substitutive reasoning all the time in math. You will get used to doing this with functional programs, too!

6 Parallelism

When we evaluate a functional program fragment, the order of evaluation of independent sub-expressions does not affect the final result. For example, the value of \((2+4)*(3+4)\) is 42, and it doesn’t matter if we calculate the value using left-right evaluation, or right-left evaluation, or parallel evaluation of the two sub-expressions. In mathematical notation we would all agree that:

\[
(2+4)*(3+4) = 6*(3+4) = 6*7 = 42 \\
(2+4)*(3+4) = (2+4)*7 = 6*7 = 42 \\
(2+4)*(3+4) = 6*7 = 42.
\]

Addition on the integers is an associative operation, i.e. \(x + (y + z) = (x + y) + z\), for all integers \(x, y, z\). Hence it should be fairly obvious that in summing the integers in a collection of integer lists, we should be able to add up the entries in each list independently of all the other lists in the collection. Later in the course we will explore the use of data structures (such as trees, and sequences) that support parallel evaluation of collections of expressions. We will talk about ways to estimate the asymptotic runtime and space usage of a functional program, and we will see that parallelism can often yield more efficient code.

If we have \(n\) integers in a single list and we add them using sequentially (like in \texttt{sum}), it obviously takes \(O(n)\) time, provided we make
the natural assumption that a single addition takes constant time. There’s no obvious way to exploit parallelism if we just have a single list of integers like this and we can only access its items from the front of the list. So sequential summation of a list of length $n$ has to do $O(n)$ “work”; further, even the smartest parallel implementation is hampered by the list structure and must also do $O(n)$ additions in a row, so we say that \texttt{sum} has “span” $O(n)$.

Similarly, if we have $n$ integers held in a list of rows, each row being an integer list, using the \texttt{count} function to add all these integers (sequentially) is going to take time proportional to $n$, regardless of the lengths of the rows.

If we have $n$ integers in a list of $k$ rows, each row of length $n/k$, and an unlimited supply of processors, we could in principle\textsuperscript{1} add the rows in parallel (using \texttt{sum} for each row, taking time proportional to $n/k$) and then add the row sums (again using \texttt{sum}, in time proportional to $k$). The total runtime for this algorithm would be $O(k+n/k)$. The work here is still $O(n)$, because you have to do $n$ additions altogether. The “span” is the length of the critical path, which in this case is $O(k+n/k)$: No matter how smart you are at dividing the work among processors, you have to wait until the first phase is over (time $n/k$) before starting the second phase (another $k$ units of time).

Finally, if we had $n$ integers at the leaves of a balanced binary tree and an unlimited supply of processors, we could use a parallel divide-and-conquer strategy, starting at the leaves and working up towards the root in $O(\log n)$ phases, each phase taking $O(1)$ time. The work here is $O(n)$ again, but the span is $O(\log n)$.

Here we have been rather informal about the precise meanings of the terms “work” and “span”. Later we will develop these ideas in more detail. Throughout the course we will pay attention to the work and span of the code that we develop. This should help you to become familiar with the potential benefits of parallelism, and you should develop an appreciation for whether and where you can safely exploit parallel evaluation and you should be able to figure out what asymptotic benefits this can bring.

\textsuperscript{1}Assuming we have primitive functions \texttt{reduce} and \texttt{map} that use parallel evaluation, we might express this algorithm as \texttt{fun parcount R = reduce (op +) (map sum R)}. Such “higher-order” functions permit very concise and elegant program designs. We will return to this topic later!
7 Self-test 1

These questions test your understanding of the lecture slides and these notes. We haven’t yet gone into the details behind the notation, so use your intuition and common sense to fill the gaps. Try to understand the reason behind each answer. (No credit, but it’s still a good idea to do this self-test!)

1. What values belong to the following types? In each case, describe the set of values and give an ML expression having the given type. For example, for the type int * real the values are pairs of an integer and a real number; for instance the ML expression (0.12,42) is a (syntactic) value of this type.
   
   (a) real * int
   (b) real -> real
   (c) int list -> int
   (d) int -> int list

2. Which of the following assertions about equivalence, if any, are true?
   
   (a) fn x:int => (21+21) is equivalent to fn x:int => 42
   (b) fn x:int => 42 is equivalent to fn y:int => 42
   (c) (fn x:int => (21+21))(3+3) = 42
   (d) (fn x:int => (21+21))(6) = (fn y:int => 42)(3+3)

3. Write a recursive ML function product of type int list -> int such that for all integer lists L, product L evaluates to the product of the items in L. The product of the empty list of integers is defined to be 1.

4. Which, if any, of the following are true?
   
   (a) product [1,2,3] = 6
   (b) product [1,2,3] = product [3,2,1]
   (c) product [0,42,42,42,42] = product [42,42,42,42,0]
5. Which, if any, of the following are true?
(a) product [1,2,3] =>* 6
(b) product [1,2,3] =>* product [3,2,1]
(c) product [0,42,42,42,42] =>* 0
(d) product [42,42,42,42,0] =>* 0

6. Investigate the different forms of syntax allowed in ML for real numbers. As you have seen already, 0.12 is legal syntax. You can also about use E-notation, such as 2E10. Get familiar with the notation, what it means, and what happens when you add or multiply. This is really an open-ended exploration, but will help later when we solve some non-trivial problems concerning real numbers.

7. We can represent a point as a pair (x,y) of type real*real. Two points whose x-coordinates are unequal determine a straight line: there is a unique line that passes through the points. A line has a slope and a y-intercept, both of which are real numbers. The line with slope a and y-intercept b is often written in math notation as “the line y = ax + b”. The line determined by (x1,y1) and (x2,y2) has slope (y2 - y1)/(x2 - x1) and y-intercept (y2x1 - x2y1)/(x2 - x1).

Write an ML function based on these ideas, by filling in this template:

```
fun line ((x1:real,y1:real), (x2:real,y2:real)) : (real -> real) =
let
    ....
    val a = ...
    val b = ...
    in
    fn (x:real) => a * x + b
    end
```

Use the first .... to insert any useful declaration you need to avoid duplication of effort!

What are the (syntactic) values of the following expressions?
(a) line ((0.0,1.0), (1.0,2.0))
(b) line ((0.0,1.0), (1.0,2.0)) 3.0
8. To convert a temperature, a real number of degrees, from Fahrenheit to Celsius: subtract 32.0 then multiply by 5.0/9.0. To convert from Celsius to Fahrenheit, multiply by 9.0/5.0 then add 32.0. For example, 212.0 degrees Fahrenheit converts to 100.0 Celsius, and 100.0 degrees Celsius converts back to 212.0 degrees Fahrenheit. Write ML functions
   
   c_to_f : real -> real
   f_to_c : real -> real

   that implement these conversions: e.g. c_to_f goes from Celsius to Fahrenheit. What are the values of:

   (i) f_to_c 451.0
   (ii) c_to_f (~273.15)
   (iii) f_to_c (~40.0)

9. The greatest common divisor (g.c.d.) of two positive integers \( x \) and \( y \) is defined to be the largest integer \( z \) such that \( z \) divides \( x \) and \( z \) divides \( y \), i.e. \( x \mod z = 0 \) and \( y \mod z = 0 \). Explain why every pair of positive integers has a g.c.d. (This is a math issue, not programming — yet.)

10. Here is a recursive function for computing the g.c.d. of positive integers:

    fun gcd(x:int, y:int):int =
        if x>y then gcd(x-y, y) else
        if y>x then gcd(x, y-x) else x

    Answer these questions by first running ML (what does the interpreter say?) and then explaining the order in which the function calls happen. Example: for \( \text{gcd}(2,3) \) ML says the value is 1:int, and

    \( \text{gcd}(2,3) \) calls \( \text{gcd}(2,1) \), which calls \( \text{gcd}(1,1) \), returns 1.

    (i) What is the value of \( \text{gcd}(24,15) \)?
    (ii) What is the value of \( \text{gcd}(\text{gcd}(24,15), \text{gcd}(2,3)) \)?
    (iii) What about \( \text{gcd}(1000000,1) \) and \( \text{gcd}(-1,42) \) and \( \text{gcd}(0,42) \)?

    Redefine \( \text{gcd} \) so that it returns 0 if either of the arguments is 0.
8 Comments

It's a bit early to be preaching about *style*, as we haven't really covered
enough of the ML language for you to be able to make an informed and
tasteful choice of syntax. And program style is a highly contentious
topic, so that one person's elegant one-liner can be viewed as horribly
obscure by others. Nevertheless in later weeks we will return to some of
the example programs introduced in the above Self Test, and see how
we could have expressed the same algorithmic content in alternative
syntax, sometimes with better taste.

For example, the `gcd` function was defined with a nested if-then-
else:

```plaintext
    if x>y then ... else if y>x ...
```

The `else` branch is itself an if-then-else expression. Some people get
all upset when you do that, and admittedly it is very easy to write
weirdly nested code that's hard to unravel. (The `gcd` function is ac-
tually acceptable to me exactly as written, since the flow of control
is so clear and simple.) But really the function needs to do one of
three things, depending on the result of comparing the values of `x`
and `y` (which will either be “less-than”, or “greater-than”, or equal). ML
provides a built-in function

```plaintext
    Int.compare : int * int -> order
```

where `order` is a built-in type. The values of type `order` are

- LESS, EQUAL, and GREATER

and can be used to indicate the result of such an integer comparison.

Here is a version of `gcd` written with a `case` expression rather than
nested if-then-else:

```plaintext
    fun gcd(x:int, y:int) : int =
      case Int.compare(x, y) of
        LESS  => gcd(x, y-x)
        | GREATER => gcd(x-y, y)
        | EQUAL   => x
```

Note also that layout on the page is worth paying attention to. We
aligned the `=>` symbols in the three clauses of the `case` expression, to
make it easier to understand the syntactic structure. Do NOT write
something as badly laid out as
fun gcd(x:int, y:int) : int = 
  case Int.compare(x, y) of LESS => gcd(x, y-x) |
    GREATER => gcd(x-y, y) |
    EQUAL    => x

A common stylistic mistake that also may signal a failure to think things through before writing your code is to use redundant syntactic structure: write something long or verbose where there’s a more succinct and clear way. For example:

fun both(x:bool, y:bool) :bool = 
  if x then (if y then true else false) else false

This is better expressed (equivalently!) as

fun both(x:bool, y:bool) :bool = 
  if x then y else false

and even more succinctly as

fun both(x:bool, y:bool) :bool = (x andalso y)

Never write if <boolean> then true else false, as it would be better to just use <boolean>. (Here <boolean> can be any expression of type bool.) Similarly

if <boolean> then false else true

is just a long-winded way of saying not <boolean>.

Another common detrimental plan is to introduce (and give names to) all sorts of helper functions. Sometimes it’s a good idea to use one, but it isn’t always. We will try to teach you to make sensible helpers and avoid irrelevant ones.

Style Advice

Study the syntax and style used in class and in lecture slides and notes. This is (usually!) chosen to illustrate and demonstrate taste and clarity of structure. Don’t use exotic pieces of ML syntax that you pick up on the internet or elsewhere, especially ones that have NOT yet been shown in class or labs. If you have questions, ask the professor or a TA! We’re always (usually) willing to pontificate.
9 Answers to Self-test 1

1. What values belong to the following types? In each case, describe the set of values and give an ML expression having the given type.

(a) real * int
Pairs of a real number (first) and an integer (second).
Examples: (1.2, 12) and (0.0, 12-12)

(b) real -> real
Functions from real numbers to real numbers.
Examples: (fn x:real => x+x) and (fn x:real => x+1.2).
The parentheses are not really needed, just kept in to aid the eye.

(c) int list -> int
Functions from integer lists to integers.
Examples: length (built-in), sum (from class), and fn L:int list => 42

(d) int -> int list
Functions from integers to integer lists.
Examples: fn x:int => [x] and fn y:int => [y,y,y]

2. Which of the following assertions about equivalence, if any, are true?

(a) fn x:int => (21+21) is equivalent to fn x:int => 42
(b) fn x:int => 42 is equivalent to fn y:int => 42
(c) (fn x:int => (21+21))(3+3) = 42
(d) (fn x:int => (21+21))(6) = (fn y:int => 42)(3+3)

ANSWER:
Each of these is true. Remember that functions are equivalent if they send equivalent arguments to equivalent results, and integer expressions are equivalent if they evaluate to the same integer. Do the calculations!
3. Write a recursive ML function `product` of type `int list -> int` such that for all integer lists `L`, `product L` evaluates to the product of the items in `L`.

The most obvious answer, similar to `sum` from class, is:

```ml
fun product [ ] = 1
  | product (x::L) = x * product L
```

By convention the product of an empty list is taken to be 1, and this makes the above function work as intended. Other solutions are also possible. However, at this early stage of the semester we expect you to write program syntax that resembles the style used in class. This is a good example, since the function definition looks very similar to the definition given in class for `sum`, using two clauses and distinguishing between the empty list and a non-empty list.

4. Which, if any, of the following are true?
   (a) `product [1,2,3] = 6`
   (b) `product [1,2,3] = product [3,2,1]`
   (c) `product [0,42,42,42,42] = product [42,42,42,42,0]`

ANSWER:
These are all true. The first is a straightforward check that `product [1,2,3]` evaluates to 6. (You do the steps!) The second one relies on the fact that integer multiplication is associative and commutative, so that `3*2*1 = 1*2*3`, and the third one follows because `0 * n = n * 0 = n` for all integers `n`. Make sure you know what associativity and commutativity mean, and check the evaluation details.

5. Which, if any, of the following are true?
   (a) `product [1,2,3] =>* 6`
   (b) `product [1,2,3] =>* product [3,2,1]`
   (c) `product [0,42,42,42,42] =>* 0`
   (d) `product [42,42,42,42,0] =>* 0`

ANSWER:
(a), (c) and (d) are true. (b) is false, because the first expression does not evaluate to the second, even though they both evaluate down to the value 6.
6. Investigate the different forms of syntax allowed in ML for real numbers. As you have seen already, 0.12 is legal syntax. You can also about use E-notation, such as 2E10. Get familiar with the notation, what it means, and what happens when you add or multiply. This is really an open-ended exploration, but will help later when we solve some non-trivial problems concerning real numbers.

ANSWER: here is an extract from an ML session (the - symbols are the system prompt):

- 2E3;
  val it = 2000.0 : real
- 2.0E3;
  val it = 2000.0 : real
- ~2E3;
  val it = ~2000.0 : real
- 2E~3;
  val it = 0.002 : real
- 2.0E3.0;
  = ;;

stdin:18.6 Error: syntax error found at DOT

As this indicates, legal syntax for reals includes \(<\text{integer}\>\text{E}\<\text{integer}\>\) and \(<\text{integer}\>.<\text{integer}\>\text{E}\<\text{integer}\>\). The value of 2E3 is the real number value of \(2 \times 10^3\), i.e. 2000.0. The expression after the E is an integer, the “exponent”. We can do arithmetic:

- 2E3 * 2E2;
  val it = 40000.0 : real
- 2E3 + 2E3;
  val it = 4000.0 : real
- 2E3 - 2E2;
  val it = 1800.0 : real
- 2E3 / 2E2;
  val it = 10.0 : real

7. We can represent a point as a pair (x,y) of type real*real. Two points whose x-coordinates are unequal determine a straight line: there is a unique line that passes through the points. A line has a slope and a y-intercept, both of which are real numbers. The line with slope a and y-intercept b is often written in math notation as “the line \(y = ax + b\)”. The line determined by \((x_1, y_1)\) and
$\langle x_2, y_2 \rangle$ has slope $(y_2 - y_1)/(x_2 - x_1)$ and y-intercept $(y_2x_1 - x_2y_1)/(x_2 - x_1)$.

Write an ML function based on these ideas, by filling in this template:

```ml
fun line ((x1:real,y1:real), (x2:real,y2:real)) : (real -> real) =
  let
    val d = (x2 - x1)
    val a = (y2 - y1) / d
    val b = (y1 * x2 - x1 * y2) / d
  in
    fn (x:real) => a * x + b
  end
```

We introduced d merely to avoid calculating $x_2 - x_1$ twice. Alternative arithmetical combinations are also possible.

What are the (syntactic) values of the following expressions?

(a) `line ((0.0,1.0), (1.0,2.0))`

Evaluates to a function value, actually:
```
fn (x:real) => let val d = (1.0 - 0.0) ... in ... end
```
ML won’t tell you the internal details.

(b) `line ((0.0,1.0), (1.0,2.0)) 3.0 4.0`

4.0

8. To convert a temperature, a real number of degrees, from Fahrenheit to Celsius: subtract 32.0 then multiply by 5.0/9.0. To convert from Celsius to Fahrenheit, multiply by 9.0/5.0 then add 32.0. For example, 212.0 degrees Fahrenheit converts to 100.0 Celsius, and 100.0 degrees Celsius converts back to 212.0 degrees Fahrenheit. Write ML functions

```ml
fun c_to_f (r:real) = (r * 1.8) + 32.0;
fun f_to_c (r:real) = (r - 32.0) * (5.0/9.0);
```

```
What are the values of:

(i) f_to_c 451.0
(ii) c_to_f (~273.15)
(iii) f_to_c (~40.0)

ANSWERS

(i) 232.777777778
(ii) ~459.67
(iii) ~40.0

9. The greatest common divisor (g.c.d.) of two positive integers \(x\) and \(y\) is defined to be the largest integer \(z\) such that \(z\) divides \(x\) and \(z\) divides \(y\), i.e. \(x \mod z = 0\) and \(y \mod z = 0\). Explain why every pair of positive integers has a g.c.d. (This is a math issue, not programming — yet.)

ANSWER: look this up in a math text or online!

10. Here is a recursive function for computing the g.c.d. of positive integers:

```ml
fun gcd (x:int, y:int):int = 
  if x>y then gcd (x-y, y) else 
  if y>x then gcd (x, y-x) else x
```

Answer these questions by first running ML (what does the interpreter say?) and then explaining the order in which the function calls happen. Example: for \(gcd(2,3)\) ML says the value is 1:int, and

\(gcd(2,3)\) calls \(gcd(2,1)\), which calls \(gcd(1,1)\), returns 1.

(i) What is the value of \(gcd(24,15)\)?

ANSWER: 3

\(gcd(24,15)\) calls \(gcd(9,15)\), which calls \(gcd(9,6)\), which calls \(gcd(3,6)\), which calls \(gcd(3,3)\), which returns 3.

(ii) What is the value of \(gcd(gcd(24,15), gcd(2,3))\)?

ANSWER

Since \(gcd(24,15)=3\) and \(gcd(2,3)=1\), the answer is the value of \(gcd(3,1)\), which is 1.

(iii) What about \(gcd(1000000,1)\) and \(gcd(-1,42)\) and \(gcd(0,42)\)?

Redefine \(gcd\) so that it returns 0 if either of the arguments is 0.

ANSWER
fun gcd (0, y:int):int = 0
  | gcd (x:int, 0) :int = 0
  | gcd (x:int, y:int) : int =
    if x>y then gcd(x-y, y) else
    if y>x then gcd(x, y-x) else x

This works, although the type annotations look awkward and repetitive, and could actually be omitted with no problems. Also this version will cause a lot of unnecessary zero-checks in recursive calls. We could avoid this issue by defining:

local
  fun loop (x:int, y:int) : int =
    if x>y then loop (x-y, y) else
    if y>x then loop (x, y-x) else x
in
  fun gcd(x:int, y:int) : int =
    if (x=0 orelse y=0) then 0 else loop (x, y)
end

We'll return to this idea — designing code to avoid extra work — later, so don't worry much about it yet.